Calcul matriciel

Nous avons déjà rencontré les matrices considérées comme des tableaux de données. Toutes les fonctions de création ou de manipulation seront reprises.

Créations de vecteurs et de matrices
eye(*n*,*p*) : matrice identité de taille *n*×*p*
matrix(*A*,*n*,*p*) : restructuration du tableau *A* selon les dimensions *n* et *p*
ones(*n*,*p*) matrice de taille *n*×*p* initialisées avec des 1
rand(*n*,*p*) matrice de taille *n*×*p* initialisées avec des valeurs aléatoires entre 0 et 1
zeros(*n*,*p*) matrice de taille *n*×*p* initialisée avec des 0

Opérations sur les matrices
Mais ces matrices sont des objets algébriques avec lesquels les règles de calcul vues en mathématiques vont être utilisées. Si *A* et *B* sont deux matrices de taille appropriée pour chaque opération, Scilab connaît les opérations suivantes :
A + B : addition matricielle. A - B : addition avec l'opposée
A*B : produit matriciel noté en mathématiques A.B
A.*B : produit élément par élément qui est propre à Scilab et qui n'est pas une opération du chapitre mathématique d'Algèbre Linéaire
A/B : division matricielle (qui n'existe pas en tant que telle en mathématiques !!!) A/B=A*B^{-1}
De même A\B qui correspond à A^{-1}*B
A./B qui est la division élément par élément. De même pour A.\B
A' est la transposée de la conjuguée et A.' est la transposée

exemple :
Si A = (1 1,-2,-2) et *B* = (1 1,-1,-1)

```
-->A*B, A.*B
 ans  =
!  0.   0. !
!  0.   0. !
 ans  =
!  1.   1. !
!  2.   2. !
```

Notons que le produit terme à terme peut très bien s'effectuer grâce à deux boucles imbriquées :

```
-->C=zeros(2,2)
 C  =
!  0.   0. !
!  0.   0. !
-->for i=1:2 do
-->for j=1:2 do
-->C(i,j)=A(i,j)*B(i,j);
-->end
-->end
-->disp(C)
!  1.   1. !
!  2.   2. !
```

Premiers exercices
1) Trouver une expression simple qui construit la matrice suivante :

```
!  1.   0.   0.   0.   0.   0.   1. !
!  0.   1.   0.   0.   0.   0.   2. !
!  0.   0.   1.   0.   0.   0.   3. !
!  0.   0.   0.   1.   0.   0.   4. !
!  0.   0.   0.   0.   1.   0.   5. !
!  0.   0.   0.   0.   0.   1.   6. !
!  1.   2.   3.   4.   5.   6.   7. !
```

2) Ecrire une fonction calculant le produit scalaire de deux vecteurs réels $\vec{u}$ et $\vec{v}$:

$$\vec{u}.\vec{v} = \sum_{i=1}^{n} u_i v_i$$

Exemple d'utilisation en algèbre linéaire :

Image d'une courbe par la rotation (vectorielle) d'angle $\frac{\pi}{4}$

```
-->x=1:30; //tableau des abscisses
-->y=sin(x); //tableau des ordonnées correspondantes
-->plot(x,y) //visualisation de la courbe initiale
```



Vous remarquerez que le graphique créé par l'instruction plot est un graphique polygonal et que le nombre de points utilisés ici n'est pas suffisant pour obtenir un aperçu "lisse".

```
-->A=[0.707 -0.707;0.707 0.707]  //matrice
```

de la rotation d'angle $\frac{\pi}{4}$

```
 A  =
!    .707  -  .707 !
!    .707     .707 !
-->for i=1:30 do
-->Z=A*[x(i) y(i)]'; //calcul de l'image de chaque point
-->zx(i)=Z(1,1); //extraction de l'abscisse de l'image
-->zy(i)=Z(2,1); //extraction de l'ordonnée de l'image
-->end
```

-->plot(zx,zy) // visualisation de la courbe image par la rotation d'angle $\frac{\pi}{4}$



Principe de la méthode du pivot de Gauss

La méthode de Gauss est une méthode permettant de résoudre les systèmes linéaires mis sous forme matricielle $A.x = b$ où A est une matrice carrée $n \times n$, x un vecteur de dimension n représentant les inconnues et b le vecteur de dimension n représentant le second membre.

$$\begin{cases} A_{1,1}x_1 + A_{1,2}x_2 + A_{1,3}x_3 = b_1, A_{2,1}x_1 + A_{2,2}x_2 + A_{2,3}x_3 = b_2 \\ A_{3,1}x_1 + A_{3,2}x_2 + A_{3,3}x_3 = b_3 \end{cases}$$

La méthode consiste à transformer le système $A.x = b$ en un autre système équivalent $T.x = c$ avec T matrice triangulaire supérieure. On obtient donc un nouveau système :

$$\begin{cases} T_{1,1}x_1 + T_{1,2}x_2 + T_{1,3}x_3 = c_1, & T_{2,2}x_2 + T_{2,3}x_3 = c_2 \\ T_{3,3}x_3 = c_3 \end{cases}$$

qui se résout très facilement en récupérant les x_i de proche en proche en commençant pas x_n.

Pour permettre de trianguler (ou trianguler) la matrice initiale, on a besoin d'effectuer des manipulations (dites élémentaires) sur A :

- échanger deux lignes de A (recherche et placement du pivot)
- multiplication d'une ligne de A par un scalaire non nul (afin de préparer la manipulation suivante)
- addition à une ligne de A une autre ligne de A

Lorsqu'à chaque étape de ces manipulations, il existe toujours un pivot non nul, la matrice (et donc le système) pourra se triangulariser. Si les mêmes manipulations sont effectuées sur le vecteur b (matrice colonne), nous pourrons alors obtenir le système équivalent noté sous forme matricielle $T.x = c$.

Remarque :
Attention, les dépassements de capacité sur les réels ne sont pas testés ; on risque des erreurs d'exécution comme le montre l'exemple suivant :

```
-->A=[1 2 3 4;9 8 7 6;5 -5 -8 3;-4 -1 -2 -3]
 A  =
!   1.    2.    3.    4. !
!   9.    8.    7.    6. !
!   5.  - 5.  - 8.    3. !
! - 4.  - 1.  - 2.  - 3. !
-->b=[1 2 3 0]'
 b  =
!   1. !
!   2. !
!   3. !
!   0. !
-->c=resoudre(A,b)
 c  =
! -  .175     .7446429  -  .7642857     .4946429 !
-->A*c'   //transformation de la solution en un vecteur colonne
 ans  =
!   1.        !
!   2.        !
!   3.        !
!   1.110E-15 !    //et 1,11.10^-15 ≠ 0
```

Exemple 1 : problème de trafic automobile.
De nombreux systèmes sont modélisés par un système linéaire, c'est-à-dire par une relation :

$$Ax = b$$

où A est une matrice (mxn) connue, b un vecteur de taille n connu et x est le vecteur que l'on cherche.

Considérons le double carrefour suivant :



Les chiffres représentent le trafic c'est-à-dire le nombre de véhicules par heure. On suppose évidemment que pour chaque carrefour toutes les voitures qui arrivent en repartent : ceci est une des lois de base des réseaux. C'est la loi dite des noeuds ou première loi de Kirchhoff. Elle est valable pour les réseaux électriques, d'eau, de télécommunications...

Dans notre cas on s'intéresse au débit en véhicules par heure et pour les 2 trafics manquants on obtient :
$x_1 = 1200 + 200$ pour la première intersection et
$x_1 + x_2 = 1600 + 700$ pour la deuxième intersection.

On a donc comme ci-dessus Ax = b avec A = [1 0; 1 1] et b = [1400 2300] .

```
-->A=[1 0;1 1]
 A  =
!   1.    0. !
!   1.    1. !
-->b=[1400 ; 2300]
 b  =
!   1400. !
!   2300. !
-->x = A\b //on cherche la solution x de A.x = b donc x = A^-1.b = A\b qui se lit "A divise b"
 x  =                                                    (A ne pas écrire ni interpréter ainsi en maths !!!)
!   1400. !
!   900.  !
```

Dans le cas présent le résultat se fait sans calcul mais son utilité peut s'imaginer lorsque A est la matrice origines-destinations d'une ville et que l'on a quelques milliers de routes et de carrefours et que l'on cherche les trafics sur les tronçons.

Exemple 2 : le zapping à la télévision

Un sondage rapide sur l'écoute des chaînes de télévision montre que lors d'une soirée pour les 4 chaînes TF1, A2, FR3, M6 et CA (autres chaînes du câble) les pourcentages de téléspectateurs sont respectivement de 30, 20, 15, 15, 20 (ces chiffres sont pure fiction et toute ressemblance avec la réalité n'est que pure coïncidence !). Chaque jour, au cours de la soirée les téléspectateurs zappent d'une chaîne vers les autres avec les pourcentages suivants :

70	15	10	10	5
10	70	10	5	5
5	5	60	5	5
5	5	10	70	5
10	5	10	10	80

Précisons ce que signifie le tableau : sur la première colonne figurent les pourcentages de ``zapping'' des téléspectateurs de TF1. Il y a donc chaque jour 70% de ces téléspectateurs qui restent sur TF1, 10% qui passent sur A2, 5% sur FR3, 5% sur M6 et 10% vers les autres chaînes. Les colonnes suivantes correspondent au ``zapping'' des téléspectateurs des autres chaînes. La diagonale du tableau, c'est-à-dire 70 70 60 70 80 représentent donc la fidélité à la chaîne.

On souhaite connaître l'évolution du nombre de téléspectateurs par chaîne avec ces informations. Plus précisément on souhaite savoir si la répartition entre les chaînes va se stabiliser et si oui quelle sera la répartition d'équilibre. Le vecteurs des pourcentages initiaux est :

```
-->p
 p  =
 !  30. 20. 15. 15. 20. !
```

et la matrice qui représente les pourcentages de "zapping" des téléspectateurs de TF1 est:

```
-->A
 A  =
 !  0.7    0.15   0.1   0.1    0.05 !
!  0.1    0.7    0.1   0.05   0.05 !
!  0.05   0.05   0.6   0.05   0.05 !
!  0.05   0.05   0.1   0.7    0.05 !
!  0.1    0.05   0.1   0.1    0.8  !
-->A*p' //transformation de p en vecteur colonne grâce à p'
 ans  =
!  28.   !
!  20.25 !
!  13.25 !
!  15.5  !
!  23.   !
```

Au bout d'un jour les proportions par chaîne sont donc données par p(1)=A p ; puis le jour suivant par p(2)=A p(1) et ainsi de suite et finalement la répartition stable est telle que Ap(s)=p(s) soit (A- I)p(s)=0 (avec la contrainte de somme égale à 100 pour les termes de p).

La répartition stable est telle que les téléspectateurs qui quittent une chaîne sont compensés par ceux des autres chaînes qui arrivent.

La première méthode consiste donc à trouver la solution à notre problème en itérant la formule $p_{i+1} = A\, p_i$ soit $p_n = A^n\, p_0$:

```
-->(A^30)*p     //A*A*A*...*A*p
 ans  =
 !   23.640942 !
!   19.25045  !
!   11.111111 !
!   15.873018 !
!   30.124479 !
```

l'opérateur ``élévation à la puissance'' noté ^ s'applique donc à une matrice et il est donc facile de trouver la répartition après 30 jours (on va voir avec le résultat ci-dessous que pour 30 on a la répartition stable avec beaucoup de précision).

L'autre méthode consiste à résoudre (avec $B=A-I$), $B.p=0$. Ceci correspond à trouver le noyau de B qui est donné par la commande kernel

```
-->B=A-eye(A) //on fabrique A-I
 B  =
! - 0.3     0.15    0.1    0.1     0.05 !
!   0.1   - 0.3     0.1    0.05    0.05 !
!   0.05    0.05  - 0.4    0.05    0.05 !
!   0.05    0.05    0.1  - 0.3     0.05 !
!   0.1     0.05    0.1    0.1   - 0.2  !

-->pp=kernel(B) //le noyau de B est l'ensemble des vecteurs proportionnels à la solution
 pp  =
!   0.5026089 !
!   0.4092673 !
!   0.2362262 !
!   0.3374660 !
!   0.6404674 !
-->pp=100*pp/sum(pp) //on transforme la solution en pourcentages ; un représentant de ce noyau suffit.
 pp  =
!   23.640662 !
!   19.250253 !
!   11.111111 !
!   15.873016 !
!   30.124958 !
-->B*pp //on vérifie que ce produit est bien zero
 ans  =
   1.0E-14 *
!   0.0990852 !
!   0.1425075 !
!   0.0710586 !
!   ¯.2032012 !
!   0.1138846 !
```

Cet exemple montre quelques manipulations simples et performantes de Scilab avec du calcul vectoriel. Cet exemple simple montre cependant un point important (et bien connu !) : pour une chaîne de télévision le point dominant est pour le long terme le pourcentage de fidélité des téléspectateurs plutôt que la performance absolue instantanée.

Quelques applications du calcul matriciel

Application 1

Matrice associée à un graphe

Le diagramme fléché ci-contre représente des chemins reliant p quatre points entre eux. Ainsi, par exemple, la flèche de P_1 vers P_2 indique qu'il est possible d'aller directement de P_1 à P_2, mais pas de P_2 à P_1



On peut représenter ce diagramme par la matrice 4x4 construite de la manière suivante : le terme de la i^e ligne et de la j^e colonne est égal à 0 s'il n'y a pas de flèche de P_i vers P_j, ce terme est égal à 1 s'il y a une flèche de P_i vers P_j.

1) Construire la matrice M représentant le diagramme ci-contre.

Une interprétation de M^2

On note N la matrice 4x4 obtenue de la manière suivante : le nombre figurant dans la i^e ligne et dans la j^e colonne est égal au nombre de chemins de longueur 2 reliant P_i à P_j.

Ainsi par exemple le coefficient se trouvant dans la première ligne et la quatrième colonne est égal à 2 car il y a deux chemins de longueur 2 reliant P_1 à P_4 : $P_1P_2P_4$ et $P_1P_3P_4$

1) Explicitez la matrice N.

2) Vérifiez que $M.M = N$.

Un autre exemple

On considère le diagramme fléché ci-contre.

Déterminer le nombre de chemins de longueur 5 allant de P_1 à P_2.



Application 2

Chaînes de Markov

On imagine deux îles voisines isolées du reste du monde qui n'ont d'échanges de populations qu'entre elles. On suppose que, d'une année à l'autre, l'île *Azur* conserve 80% de sa population et accueille 20% des habitants de l'île *Beauté* (mariages, par exemple).

On suppose aussi la stabilité de cet échange pendant un certain nombre d'années et, sur chaque île, le solde naissance-décès est nul.

1) On note $p_n = \begin{pmatrix} a_n \\ b_n \end{pmatrix}$ le vecteur colonne où a_n désigne la population de l'île *Azur* au début de l'année 2000 + n et b_n désigne la population de l'île *Beauté* au début de l'année 2000 + n.

Exprimer a_{n+1} et b_{n+1} en fonction de a_n et b_n.

En déduire la matrice A telle que $p_{n+1} = Ap_n$. (A est dite matrice de transition).

2) Au début de l'année 2000, *Azur* a 2000 habitants et *Beauté* 1000 habitants.

Estimer les populations respectives de *Azur* et *Beauté* au début de 2005 et au début de 2025.

3) Reprendre les questions du 2) avec les hypothèses suivantes sur a_0 et b_0 : $a_0 = 1800$ et $b_0 = 1200$.

Application 3

Une certaine espèce de coléoptères se comporte de la manière suivante :

En moyenne, la moitié des coléoptères meurent la première année, les autres survivent la deuxième année. Parmi ces derniers, les deux tiers meurent durant cette seconde année et les autres survivent une troisième année. À la fin de cette troisième année, ils meurent tous, mais auparavant chacun d'eux donne naissance en moyenne à six coléoptères.

1) Ecrire la matrice de transition de l'évolution de cette population.

Quelle propriété possède cette matrice ?

2) Partant d'une population de 9000 coléoptères, 3000 dans chaque tranche d'âge, donnez la composition de la population au bout de : un an ; deux ans ; trois ans. À votre avis, comment cette population va-t-elle évoluer ?

3) Prenez la répartition par tranche d'âge que vous voulez pour une population de ces insectes et donnez son évolution dans le temps.

4) Que se passe-t-il si la répartition de la population est (1800 900 300) ?

Même question avec (36 18 6).

5) On recherche s'il existe une répartition de population invariante d'une époque à la suivante. Quel système d'équations faut-il résoudre ?

D'après *L'algèbre linéaire par ses applications,* T.J. Fletcher, Éd. CÉDIC

Application 4

Dans une usine agroalimentaire, on utilise trois céréales C_1, C_2 et C_3 pour fabriquer trois farines F_1, F_2 et F_3. Dans la matrice $A = \begin{pmatrix} 50,75,45 & 25,15,20 & 25,10,35 \end{pmatrix}$, l'élément a_{ij} représente la quantité de céréales C_i, exprimée en kg, nécessaire pour fabriquer 100 kg de farine F_j.
1) Quelles quantités respectives, exprimées en tonnes, de farine F_1, F_2 et F_3 pourra-t-on fabriquer en utilisant complètement 345 tonnes de C_1, 102,5 de C_2 et 102,5 de C_3 ?
2) Quelles quantités respectives de céréales seront nécessaires pour fabriquer 2 t de farine F_1, 500 kg de F_2 et 800 kg de F_3.

Application 5
Chercher, s'il existe, une parabole d'équation $y = ax^2 + bx + c$ passant par les trois points $A(5;102,5)$, $B(10;190)$ et $C(20;440)$
Pour la présentation de la réponse, vous construirez une chaîne de caractères représentant le polynôme solution et vous construirez une représentation graphique de ce polynôme pour vérifier la validité de cette solution.

Application 6
On suppose que trois circuits de distribution de films se partagent exclusivement les salles de cinéma françaises : circuit *Action* (*A*), circuit *Boulevard* (*B*) et circuit *Cine-club* (*C*). On suppose que; d'un mois sur l'autre, pendant au moins 1 an :
- *A* conserve 80% de son implantation, mais en perd 10% au profit de *B* et 10% au profit de *C*.
- *B* conserve 70% de son implantation, mais en perd 20% au profit de *A* et 10% au profit de *C*.
- *C* conserve 60% de son implantation, mais en perd 30% au profit de *A* et 10% au profit de *B*.

On suppose, de plus, que le parc des salles reste stable durant cette période.
Si, au départ de l'étude, les parts d'implantation sont respectivement de 20% pour *A*, 50% pour *B* et 30% pour *C*, que sont-elles devenues :
au bout de 3 mois ? de 6 mois ? d'un an ?
Les parts d'implantation finales dépendent-elles de la situation initiale ?

Application 7
Le problème du plus court chemin : Etant donné un graphe orienté et valué (tel que chaque arc possède une valeur). Quel est le plus court chemin permettant d'aller du sommet *i* au sommet *j* ? (où la longueur d'un chemin est la somme des valeurs associées à chaque arc).
Algorithme de Floyd (1962) : On part de la matrice $A^{(0)}$ caractérisant le graphe qui est défini de la manière suivante : $A = (a_{ij})_{1 \leqslant i,j \leqslant n}$ où les coefficients a_{ij} sont définis par
$\begin{cases} a_{ij} = \text{la valeur de l'arc } (i,j) \text{ s'il existe}, a_{ij} = \infty \text{ si l'arc } (i,j) \text{ n'existe pas} \\ a_{ii} = 0 \end{cases}$
On calcule de façon itérative la matrice $A^{(k)} = (a^{(k)}_{ij})$ où le coefficient $a^{(k)}_{ij}$ représente la valeur du plus court chemin de *i* à *j* n'utilisant que les *k* premiers sommets. Ainsi, par exemple; $a^{(1)}_{46}$ représente le plus court chemin de 4 à 6 n'utilisant éventuellement que le sommet 1.
La solution du problème est alors fournie par $A^{(n)}$ puisque cette matrice sera celle des plus courts chemins utilisant tous les sommets.
Les coefficients de $A^{(k)}$ se calculent à partir de ceux de $A^{(k-1)}$ avec la formule suivante :

$$a^{(k)}_{ij} = \min(a^{(k-1)}_{ij}, a^{(k-1)}_{ik} + a^{(k-1)}_{kj})$$

Cette formule peut se comprendre de la façon suivante :
Le plus court chemin pour aller de *i* à *j* en n'utilisant que les sommets de 1 à *k* est le plus petit entre :
- le plus court chemin pour aller de *i* à *j* en n'utilisant que les sommets de 1 à *k* - 1 et entre
- le plus court chemin pour aller de *i* à *j* en passant par *k* et en n'utilisant que les sommets de 1 à *k* - 1 pour aller de *i* à *k* puis de *k* à *j*.

Déterminer la matrice des plus courts chemins $A^{(3)}$ du graphe orienté valué ci-contre :



```
function application1
//construction de M
M=[0 1 1 0;0 0 0 1;0 1 0 1;1 0 0 0]; disp(M,'M')
//construction de N
N=[0 1 0 2;1 0 0 0;1 0 0 1;0 1 1 0];disp(N,'N')
//Vérification M.M=N
disp(M*M,'M.M')
//autre exemple
A=[0 0 1 1 0;1 0 0 0 0;0 1 0 0 1;0 1 0 0 0;0 0 0 1 0]
B=A^5;disp(B)
disp(B(1,2),'Le nombre de chemins de P1 à P2 est')

function application2
A=[0.8 0.2;0.2 0.8]
disp(A)
p=[2000;1000]
disp(p)
for i=1:5 do p=A*p; end
disp(p)
//ou A^5*p
for i=1:25 do p=A*p; end
disp(p)
//ou A^25*p

p=[1800;1200]
disp(p)
for i=1:5 do p=A*p; end
disp(p)
//ou A^5*p
for i=1:25 do p=A*p; end
disp(p)
//ou A^25*p


function application3
A=[0 0 6;1/2 0 0;0 1/3 0]
disp(A)
p=[3000 3000 3000]'
//évolution de 3 ans
for i=1:3 do p=A*p; disp(p), end
//vérification du caractère cyclique
for i=1:3 do p=A*p; disp(p), end
disp('Nouvelles conditions initiales')
p=[1800 900 300]';disp(p)
//évolution de 3 ans
for i=1:3 do p=A*p; disp(p), end
disp('Nouvelles conditions initiales')
p=[36 18 6]';disp(p)
//évolution de 3 ans
for i=1:3 do p=A*p; disp(p), end
//invariante si de la forme p=[k k/2 k/6]

function application4
A=[50 75 45;25 15 20;25 10 35]
disp(A)
//si on pose p=[345 102.5 102.5]', il faut résoudre AX=p
p=[345000 102500 102500]'
x=A\p
disp(x*100/1000)
//x*(50C1+25C2+25C3 pour x*100kg de F1 ...

function application5
A=[25 5 1;100 10 1;400 20 1]; disp(A)
Y=[102.5;190;440];disp(Y)
X=A\Y;disp(X)
disp('y='+string(X(1,1))+'x²+'+string(X(2,1))+'x+'+string
(X(3,1)))
A=0:20;
B=X(1,1)*A.*A+X(2,1)*A+X(3,1)
plot(A,B)

function application6
A=[0.8 0.2 0.3;0.1 0.7 0.1;0.1 0.1 0.6];disp(A)
p=[20 50 30]';disp(p)
B=A^3;disp(B*p)
B=A^6;disp(B*p)
B=A^12;disp(B*p)

function application7
A=[0 8 5;3 0 %inf;%inf 2 0];disp(A)
for n=1:3 do
B=A
for i=1:3 do
for j=1:3 do
B(i,j)=min(A(i,j),A(i,n)+A(n,j))
end
end
A=B
disp(A)
end
```